
Supplemental Models for Example 2 in Mplus 
TITLE: Model 2c: Fixed Effects of Sex (0=M,1=W) by PMC Negative Mood; 

DATA:  FILE = Example2.csv;  ! Can just list file if in same folder; 

VARIABLE: 

! List of ALL variables in stacked data file, in order; 

! Mplus does NOT know what they used to be called, though; 

  NAMES ARE ID lGlucAM TVnm0 WPnm PMnm0 sexmW; 

! List of ALL variables used in model (DEFINED variables at end); 

  USEVARIABLES ARE lGlucAM WPnm PMnm0 sexMW PMnmsex; 

! Missing data codes (here, -999); 

  MISSING ARE ALL (-999); 

! Identify level-2 ID; 

  CLUSTER = ID; 

! Predictor variables with variation ONLY at level 1; 

  WITHIN = WPnm;            

! Predictor variables with variation ONLY at level 2; 

  BETWEEN = PMnm0 sexMW PMnmsex; 

DEFINE:     PMnmsex = PMnm0*sexMW;   ! Create observed level-2 interaction; 

! 2-level model with random slopes via ML (or MLR for non-normality); 

ANALYSIS:   TYPE IS TWOLEVEL RANDOM; ESTIMATOR IS ML;  

         

MODEL:   

! Level-1, Within-Person (WP) Model; 

%WITHIN%                         

 lGlucAM;                    ! L1 R: residual variance in Y; 

 WPmood | lGlucAM ON WPnm;   ! Placeholder for L1 WP mood->Y; 

 

! Level-2, Person-Level Model;  

%BETWEEN% 

[lGlucAM] (fint);      ! Fixed intercept for Y; 

 lGlucAM;              ! L2 random intercept variance in Y; 

[WPmood]  (fWPmood);          ! L1 WP fixed effect (label) of mood->Y; 

 WPmood@0;                    ! L2 G: No random mood slope variance->Y; 

 WPmood ON sexMW (fWPmsex);   ! Cross-level fixed effect of WP*women->Y;  

 lGlucAM ON PMnm0   (fBPmood);    ! BP fixed effect of mood->Y; 

 lGlucAM ON sexMW   (fsex);       ! BP total fixed effect of women->Y; 

 lGlucAM ON PMnmsex (fBPmsex);    ! L2 interaction of BP*women->Y; 

 

! Request all effects in same place for easier interpretation; 

! Those with only one term are already given by the model output; 

MODEL CONSTRAINT:                ! Equivalent to ESTIMATE in SAS; 

! Need to name each new created fixed effect; 

NEW(intmen intwom intdif WPmen WPwom WPdif  

    BPmen BPwom BPdif Contmen Contwom Contdif);  

intman = fint;                ! Intercept: Men (Mood=0);                   

intwom = fint + fsex;         ! Intercept: Women (Mood=0); 

intdif = fsex;                ! Intercept: Difference (Mood=0); 

WPmen = fWPmood;                  ! WP Effect: Men; 

WPwom = fWPmood + fWPmsex;        ! WP Effect: Women; 

WPdif = fWPmsex;                  ! WP Effect: Difference; 

BPmen = fBPmood;                      ! BP Effect: Men; 

BPwom = fBPmood + fBPmsex;            ! BP Effect: Women; 

BPdif = fBPmsex;                      ! BP Effect Difference; 

Contmen = fBPmood - fWPmood;                     ! Context Effect: Men; 

Contwom = fBPmood - fWPmood + fBPmsex - fWPmsex; ! Context Effect: Women; 

Contdif = fBPmsex - fWPmsex;                ! Context Effect: Difference; 

MODEL FIT INFORMATION 

 

Number of Free Parameters                        8 

 

Loglikelihood 

          H0 Value                         994.308 

 

Information Criteria 

          Akaike (AIC)                   -1972.615 

          Bayesian (BIC)                 -1921.988 

          Sample-Size Adjusted BIC       -1947.408 

            (n* = (n + 2) / 24) 

 

MODEL RESULTS 

                                                   Two-Tailed 

                    Estimate       S.E.  Est./S.E.    P-Value 

Within Level 

 

 Residual Variances 

    LGLUCAM            0.030      0.001     44.267      0.000 

 

Between Level 

 

 WPMOOD     ON 

    SEXMW             -0.035      0.008     -4.383      0.000 

 

 LGLUCAM    ON 

    PMNM0              0.200      0.048      4.119      0.000 

    SEXMW             -0.036      0.036     -0.999      0.318 

    PMNMSEX           -0.185      0.061     -3.017      0.003 

 

 Intercepts 

    LGLUCAM            4.954      0.027    181.203      0.000 

    WPMOOD             0.031      0.006      5.174      0.000 

 

 Residual Variances 

    LGLUCAM            0.061      0.006      9.928      0.000 

    WPMOOD             0.000      0.000    999.000    999.000 

 

BOLD VALUES ARE NOT GIVEN DIRECTLY BY THE MODEL 

 

New/Additional Parameters 

    INTMEN             4.954      0.027    181.203      0.000 

    INTWOM             4.918      0.024    206.412      0.000 

    INTDIF            -0.036      0.036     -0.999      0.318 

    WPMEN              0.031      0.006      5.174      0.000 

    WPWOM             -0.003      0.005     -0.648      0.517 

    WPDIF             -0.035      0.008     -4.383      0.000 

    BPMEN              0.200      0.048      4.119      0.000 

    BPWOM              0.015      0.038      0.389      0.697 

    BPDIF             -0.185      0.061     -3.017      0.003 

    CONTMEN            0.168      0.049      3.447      0.001 

    CONTWOM            0.018      0.038      0.473      0.637 

    CONTDIF           -0.151      0.062     -2.434      0.015 

TITLE: Model 3c: Fixed Effects of Sex (0=M,1=W) by GMC Negative Mood; 

( DATA, DEFINE, and ANALYSIS are the same ) 

 VARIABLE: 

! List of ALL variables in stacked data file, in order; 

! Mplus does NOT know what they used to be called, though; 

  NAMES ARE ID lGlucAM TVnm0 WPnm PMnm0 sexmW; 

! List of ALL variables used in model (DEFINED variables at end); 

  USEVARIABLES ARE lGlucAM TVnm0 PMnm0 sexMW PMnmsex; 

! Missing data codes (here, -999); 

  MISSING ARE ALL (-999); 

! Identify level-2 ID; 

  CLUSTER = ID; 

! Predictor variables with variation ONLY at level 1; 

  WITHIN = TVnm0;            

! Predictor variables with variation ONLY at level 2; 

  BETWEEN = PMnm0 sexMW PMnmsex; 

         

MODEL:   

! Level-1, Within-Person (WP) Model; 

%WITHIN%                         

 lGlucAM;                     ! L1 R: residual variance in Y; 

 WPmood | lGlucAM ON TVnm0;   ! Placeholder for L1 WP mood->Y; 

 

! Level-2, Person-Level Model;  

%BETWEEN% 

[lGlucAM] (fint);      ! Fixed intercept for Y; 

 lGlucAM;              ! L2 random intercept variance in Y; 

[WPmood]  (fWPmood);         ! L1 WP fixed effect (label) of mood->Y; 

 WPmood@0;                   ! L2 G: No random mood slope variance->Y; 

 WPmood ON sexMW  (fWPmsex); ! Cross-level fixed effect of WP*women->Y;  

 lGlucAM ON PMnm0 (fconmood);   ! BP contextual effect of mood->Y; 

 lGlucAM ON sexMW (fsex);       ! BP total fixed effect of women->Y; 

 lGlucAM ON PMnmsex (fconsex);  ! L2 interaction of contextual*women->Y; 

 

! Request all effects in same place for easier interpretation; 

! Those with only one term are already given by the model output; 

MODEL CONSTRAINT:                ! Equivalent to ESTIMATE in SAS; 

! Need to name each new created fixed effect; 

NEW(intmen intwom intdif WPmen WPwom WPdif  

    BPmen BPwom BPdif Contmen Contwom Contdif);  

intmen = fint;               ! Intercept: Men (Mood=0);                   

intwom = fint + fsex;        ! Intercept: Women (Mood=0); 

intdif = fsex;               ! Intercept: Difference (Mood=0); 

WPmen = fWPmood;                  ! WP Effect: Men; 

WPwom = fWPmood + fWPmsex;        ! WP Effect: Women; 

WPdif = fWPmsex;                  ! WP Effect: Difference; 

BPmen = fWPmood + fconmood;                       ! BP Effect: Men 

BPwom = fWPmood + fconmood + fWPmsex + fconsex;   ! BP Effect: Women 

BPdif = fWPmsex + fconsex;                        ! BP Effect Difference 

Contmen = fconmood;                     ! Contextual Effect: Men 

Contwom = fconmood + fconsex;           ! Contextual Effect: Women 

Contdif = fconsex;                      ! Contextual Effect: Difference 

 

! Now the BP contextual effects are given directly instead of the BP total effects; 

MODEL FIT INFORMATION – THIS MODEL IS EQUIVALENT TO 2C 

 

Number of Free Parameters                        8 

 

Loglikelihood 

          H0 Value                         994.308 

 

Information Criteria 

          Akaike (AIC)                   -1972.615 

          Bayesian (BIC)                 -1921.988 

          Sample-Size Adjusted BIC       -1947.408 

            (n* = (n + 2) / 24) 

 

MODEL RESULTS 

                                                   Two-Tailed 

                    Estimate       S.E.  Est./S.E.    P-Value 

 

Within Level 

 

 Residual Variances 

    LGLUCAM            0.030      0.001     44.267      0.000 

 

Between Level 

 

 WPMOOD     ON 

    SEXMW             -0.035      0.008     -4.383      0.000 

 

                      BOLD VALUES ARE DIFFERENT IN THIS MODEL 

 LGLUCAM    ON 

    PMNM0              0.169      0.049      3.450      0.001       

    SEXMW             -0.036      0.036     -0.999      0.318 

    PMNMSEX           -0.150      0.062     -2.433      0.015 

 

 Intercepts 

    LGLUCAM            4.954      0.027    181.203      0.000 

    WPMOOD             0.031      0.006      5.174      0.000 

 

 Residual Variances 

    LGLUCAM            0.061      0.006      9.928      0.000 

    WPMOOD             0.000      0.000    999.000    999.000 

 

BOLD VALUES ARE NOT GIVEN DIRECTLY BY THE MODEL 

 

New/Additional Parameters 

    INTMEN             4.954      0.027    181.203      0.000 

    INTWOM             4.918      0.024    206.412      0.000 

    INTDIF            -0.036      0.036     -0.999      0.318 

    WPMEN              0.031      0.006      5.174      0.000 

    WPWOM             -0.003      0.005     -0.648      0.517 

    WPDIF             -0.035      0.008     -4.383      0.000 

    BPMEN              0.200      0.048      4.119      0.000 

    BPWOM              0.015      0.038      0.389      0.697 

    BPDIF             -0.185      0.061     -3.017      0.003 

    CONTMEN            0.168      0.049      3.447      0.001 

    CONTWOM            0.018      0.038      0.473      0.637 

    CONTDIF           -0.151      0.062     -2.434      0.015 

 


