
Supplemental Models for Example 1 in Mplus 
 

TITLE:  Model 3 via MLM:  

        Attitudes Predict Intercept, Linear, and Quadratic Age Slopes 

DATA:   FILE = Example1.csv;   ! Syntax in same folder as data 

VARIABLE: 

! List of variables in data file 

  NAMES = PersonID occasion risky age18 att4 mon3 agesq; 

! Variables to be analyzed in this model 

  USEVARIABLE = age18 agesq att4 risky; 

  MISSING ARE ALL (-999);   ! Missing data identifier 

! MLM options 

  CLUSTER = PersonID;       ! Level-2 ID 

  BETWEEN = att4;           ! Observed ONLY level-2 predictors 

  WITHIN  = age18 agesq;    ! Observed ONLY level-1 predictors 

 

ANALYSIS:   TYPE = TWOLEVEL RANDOM; ESTIMATOR = ML;  

 

MODEL:  ! R = risky behavior 

%WITHIN% 

 risky (Rresvar);                ! L1 R: Residual variance (label) 

 Rslp  | risky ON age18;         ! Placeholder for R linear age slope 

 Rquad | risky ON agesq;         ! Placeholder for R quadratic age slope 

 

%BETWEEN% 

[risky];                         ! Fixed intercept 

 risky (Rintvar);                ! L2 G: Random intercept variance (label) 

[Rslp Rquad];                    ! Fixed age slopes (as defined earlier) 

 Rslp (Rslpvar);                 ! L2 G: Random linear age slope variance 

 Rquad@0;                        ! No quadratic age slope variance 

 risky Rslp Rquad ON att4;       ! Att-> R int, linear, quad age slopes 

 risky WITH Rslp (RIntSlp);      ! R Int-slope covariance (label) 

MODEL FIT INFORMATION 

 

Number of Free Parameters                       10 

 

Loglikelihood 

          H0 Value                       -3799.718 

 

Information Criteria 

          Akaike (AIC)                    7619.437 

          Bayesian (BIC)                  7671.879 

          Sample-Size Adjusted BIC        7640.113 

            (n* = (n + 2) / 24) 

 

 

MODEL RESULTS 

 

                                                    Two-Tailed 

                    Estimate       S.E.  Est./S.E.    P-Value 

 

Within Level 

 

 Residual Variances 

    RISKY              8.324      0.372     22.353      0.000 

 

Between Level 

 

 RSLP       ON 

    ATT4              -0.900      0.242     -3.721      0.000 

 

 RQUAD      ON 

    ATT4              -0.064      0.036     -1.762      0.078 

 

 RISKY      ON 

    ATT4              -3.476      0.580     -5.989      0.000 

 

 RISKY    WITH 

    RSLP               1.884      0.356      5.286      0.000 

 

 Intercepts 

    RISKY             23.299      0.350     66.573      0.000 

    RSLP               1.948      0.146     13.336      0.000 

    RQUAD              0.142      0.022      6.461      0.000 

 

 Residual Variances 

    RISKY             18.083      2.204      8.206      0.000 

    RSLP               0.486      0.080      6.087      0.000 

    RQUAD              0.000      0.000    999.000    999.000 

TITLE:  Model 3 via SEM:  

        Attitudes Predict Intercept, Linear, and Quadratic Age Slopes 

DATA:   FILE = Example1.csv;   ! Syntax in same folder as data 

 

! Unstacking to multivariate format 

DATA LONGTOWIDE: 

! Names of old stacked former variables (without numbers) 

  LONG = risky|age;   

! Names of new multivariate variables (that use numbers) 

  WIDE = risky12-risky18|age12-age18;  

! Variable with level-2 ID info  

  IDVARIABLE = PersonID;        

! Old level-1 identifier         

  REPETITION = occasion (12 13 14 15 16 17 18); 

 

VARIABLE: 

! List of variables in original data file 

  NAMES = PersonID occasion risky age att4 mon agesq; 

! Variables to be analyzed in this model 

  USEVARIABLE = att4 risky12-risky18 age12-age18; 

  MISSING ARE ALL (-999);   ! Missing data identifier 

  TSCORES = age12-age18;    ! Exact time indicator 

 

ANALYSIS:   TYPE = RANDOM; ESTIMATOR = ML; MODEL = NOCOVARIANCES; 

MODEL:  ! R = risky behavior 

[risky12-risky18@0];         ! All variable intercepts fixed to 0 

 risky12-risky18 (Rresvar);  ! L1 R: R residual variances held equal 

 

! Risky behavior quadratic growth model using exact age as loadings 

  Rint Rslp Rquad | risky12-risky18 AT age12-age18; 

 

! Fixed growth effects 

  [Rint Rslp Rquad];             

! L2 G: Random int and linear age variances, no quad age variance 

  Rint Rslp Rquad@0;   

! L2 G: Within-variable random int-slope covariance            

  Rint WITH Rslp;     

! Attitudes --> risky int, linear slope 

  Rint Rslp Rquad ON att4;              

 

MODEL FIT INFORMATION 

 

Number of Free Parameters                       10 

 

Loglikelihood 

 

          H0 Value                       -3799.718 

 

Information Criteria 

 

          Akaike (AIC)                    7619.437 

          Bayesian (BIC)                  7652.420 

          Sample-Size Adjusted BIC        7620.739 

            (n* = (n + 2) / 24) 

 

 

 

MODEL RESULTS – INTERCEPTS HELD TO 0 OMITTED FOR BREVITY 

 

                                                    Two-Tailed 

                    Estimate       S.E.  Est./S.E.    P-Value 

 

 RINT       ON 

    ATT4              -3.476      0.580     -5.989      0.000 

 

 RSLP       ON 

    ATT4              -0.900      0.242     -3.721      0.000 

 

 RQUAD      ON 

    ATT4              -0.064      0.036     -1.762      0.078 

 

 RINT     WITH 

    RSLP               1.884      0.356      5.286      0.000 

 

 Intercepts 

    RINT              23.299      0.350     66.571      0.000 

    RSLP               1.948      0.146     13.336      0.000 

    RQUAD              0.142      0.022      6.462      0.000 

 

 Residual Variances 

    RISKY12            8.324      0.372     22.353      0.000 

    RISKY13            8.324      0.372     22.353      0.000 

    RISKY14            8.324      0.372     22.353      0.000 

    RISKY15            8.324      0.372     22.353      0.000 

    RISKY16            8.324      0.372     22.353      0.000 

    RISKY17            8.324      0.372     22.353      0.000 

    RISKY18            8.324      0.372     22.353      0.000 

    RINT              18.084      2.204      8.206      0.000 

    RSLP               0.486      0.080      6.087      0.000 

    RQUAD              0.000      0.000    999.000    999.000 

 

 


